

1. Create an associative array and populate it with the instructor's full name. In other words, each row of the associative array should contain the first name, middle initial, and last name. Display this information on the screen.

**SOLUTION:**

```
SET SERVEROUTPUT ON
DECLARE
    CURSOR instructor_name IS
        SELECT first_name||' '||last_name AS full_name
        FROM instructor;

    TYPE full_name_type IS TABLE OF instructor.first_name%TYPE
    INDEX BY BINARY_INTEGER;
    full_name_tab full_name_type;
    v_counter INTEGER := 0;

BEGIN
    FOR full_name_rec IN instructor_name LOOP
        v_counter := v_counter + 1;
        full_name_tab(v_counter) := full_name_rec.full_name;
        DBMS_OUTPUT.PUT_LINE('full_name('||v_counter||'): '||full_name_tab(v_counter));
    END LOOP;
END;
```

**OUTPUT:**

```
full_name(1): Anca Pup
full_name(2): Alexa Iuga
full_name(3): Ema Pop
full_name(4): Andrei Opra
full_name(5): Maria Costa
```

2. Modify the script you just created. Instead of using an associative array, use a varray.

**SOLUTION:**

```
SET SERVEROUTPUT ON
DECLARE
    CURSOR instructor_name IS
        SELECT first_name||' '||last_name AS full_name
        FROM instructor;

    TYPE full_name_type IS VARRAY(50) OF instructor.first_name%TYPE;
    full_name_varray full_name_type := full_name_type();
    v_counter INTEGER := 0;

BEGIN
```

```

FOR full_name_rec IN instructor_name LOOP
    v_counter := v_counter + 1;
    full_name_varray.EXTEND;
    full_name_varray(v_counter) := full_name_rec.full_name;
    DBMS_OUTPUT.PUT_LINE('full_name('||v_counter||'): '||full_name_varray(v_counter));
END LOOP;
END;

```

### OUTPUT:

```

full_name(1): Anca Pup
full_name(2): Alexa Iuga
full_name(3): Ema Pop
full_name(4): Andrei Opra
full_name(5): Maria Costa

```

3. Modify the script you just created. Create an additional varray, and populate it with unique course numbers for the courses that each instructor teaches. Display the instructor's name and the list of courses he or she teaches.

### SOLUTION:

```

SET SERVEROUTPUT ON
DECLARE
    CURSOR instructor_name IS
        SELECT instructor_id, first_name||' '||last_name AS full_name
        FROM instructor;

    CURSOR course_no_cursor(instructor_id_param NUMBER) IS
        SELECT UNIQUE course_no course_identifier
        FROM section
        WHERE instructor_id = instructor_id_param;

    TYPE full_name_type IS VARRAY(50) OF instructor.first_name%TYPE;
    full_name_varray full_name_type := full_name_type();

    TYPE course_no_type IS VARRAY(50) OF course.course_no%TYPE;
    course_no_varray course_no_type := course_no_type();

    v_counter_1 INTEGER := 0;
    v_counter_2 INTEGER := 0;

BEGIN
    FOR full_name_rec IN instructor_name LOOP
        v_counter_1 := v_counter_1 + 1;
        full_name_varray.EXTEND;
        full_name_varray(v_counter_1) := full_name_rec.full_name;

```

```

DBMS_OUTPUT.PUT_LINE('instructor: '||full_name_varray(v_counter_1)||' teaches: ');

v_counter_2 := 0;
course_no_varray := course_no_type();

FOR course_no_rec IN course_no_cursor(full_name_rec.instructor_id) LOOP
    v_counter_2 := v_counter_2 + 1;
    course_no_varray.EXTEND;
    course_no_varray(v_counter_2) := course_no_rec.course_identifier;
    DBMS_OUTPUT.PUT_LINE('          '||course_no_varray(v_counter_2));

END LOOP;

END LOOP;
END;

```

### OUTPUT:

```

instructor: Anca Pup teaches:
    11010
    11030
instructor: Alexa Iuga teaches:
    11020
    11090
    11030
instructor: Ema Pop teaches:
instructor: Andrei Opra teaches:
    11090
    11080
instructor: Maria Costa teaches:

```

#### 4. Find and explain the errors in the following script:

```

DECLARE
TYPE varray_type1 IS VARRAY(7) OF INTEGER;
TYPE table_type2 IS TABLE OF varray_type1 INDEX BY
BINARY_INTEGER;
varray1 varray_type1 := varray_type1(1, 2, 3);
table2 table_type2 := table_type2(varray1,
varray_type1(8, 9, 0));
BEGIN
DBMS_OUTPUT.PUT_LINE ('table2(1)(2): '||table2(1)(2));
FOR i IN 1..10 LOOP
varray1.EXTEND;
varray1(i) := i;
DBMS_OUTPUT.PUT_LINE ('varray1('||i||')': '||varray1(i));
END LOOP;
END;

```

SOLUTION:

Error report -

ORA-06550: line 6, column 35:  
PLS-00681: named association syntax is required  
ORA-06550: line 6, column 8:  
PL/SQL: Item ignored  
ORA-06550: line 9, column 43:  
PLS-00320: the declaration of the type of this expression is incomplete or malformed  
ORA-06550: line 9, column 1:  
PL/SQL: Statement ignored  
06550. 00000 - "line %s, column %s:\n%s"  
\*Cause: Usually a PL/SQL compilation error.

ORA-01403: no data found  
ORA-06512: at line 9  
01403. 00000 - "no data found"

Error report -  
ORA-06532: Subscript outside of limit

### **SOLUTION:**

```
DECLARE
  TYPE varray_type1 IS VARRAY(7) OF INTEGER;
  TYPE table_type2 IS TABLE OF varray_type1
  INDEX BY BINARY_INTEGER;
  varray1 varray_type1 := varray_type1(1, 2, 3);
  table2 table_type2;
BEGIN
  table2(1) := varray1;
  table2(2) := varray_type1(8, 9, 0);
  DBMS_OUTPUT.PUT_LINE ('table2(1)(2): ' || table2(1)(2));
  FOR i IN 1..4 LOOP
    varray1.EXTEND;
    varray1(i) := i;
    DBMS_OUTPUT.PUT_LINE ('varray1(' || i || '): ' || varray1(i));
  END LOOP;
END;
```

### **OUTPUT:**

```
table2(1)(2): 2
varray1(1): 1
varray1(2): 2
varray1(3): 3
varray1(4): 4
```